Práctica 2

Representación de Curvas con ordenador

2.1. Enunciado de la Práctica

En esta práctica se trata de obtener la representación gráfica de las curvas siguientes:

1. Curvas definidas como gráfica de funciones, $y = f(x)$, donde

$$C_1 \equiv \left\{ \begin{array}{lcl} y & = & 16\dfrac{x - \operatorname{sen} x}{x^2} \\ x & \in & [0, 2\pi] \end{array} \right. \quad ; \quad C_2 \equiv \left\{ \begin{array}{lcl} y & = & \dfrac{\ln x}{x^3} \\ x & \in & [0.8; 3] \end{array} \right. \quad ; \quad C_3 \equiv \left\{ \begin{array}{lcl} y & = & e^{\operatorname{sen} x} \\ x & \in & [0, 2\pi] \end{array} \right.$$

2. Curvas definidas mediante coordenadas polares, $\rho = \rho(\alpha)$, mediante las siguientes expresiones:

$$C_4 \equiv \left\{ \begin{array}{l} \rho = 3 \\ \alpha \in [0, 2\pi] \end{array} \right. \quad ; \quad C_5 \equiv \left\{ \begin{array}{l} \rho = 4 \operatorname{sen}(3\alpha) \\ \alpha \in [0, 2\pi] \end{array} \right. \quad ; \quad C_6 \equiv \left\{ \begin{array}{l} \rho = 4 + 4\cos\alpha \\ \alpha \in [0, 2\pi] \end{array} \right.$$

$$C_7 \equiv \left\{ \begin{array}{l} \rho = \dfrac{1}{1 + 0.9\cos\,\alpha} \\ \alpha \in [0, 2\pi] \end{array} \right. \quad ; \quad C_8 \equiv \left\{ \begin{array}{l} \rho(\alpha) = \sqrt{2}\sqrt{\cos 2\theta} \\ \alpha \in [0, 2\pi] \end{array} \right.$$

3. Curvas definidas mediante ecuaciones paramétricas:

$$C_9 \equiv \left\{ \begin{array}{lcl} x & = & \cos^3 t \\ y & = & \operatorname{sen}^3 t \\ t & \in & [0, 2\pi] \end{array} \right. \quad ; \quad C_{10} \equiv \left\{ \begin{array}{lcl} x & = & 4t - 4\operatorname{sen} t \\ y & = & 4 - 4\cos t \\ t & \in & [0, 4\pi] \end{array} \right.$$

4. Curvas definidas mediante ecuaciones implícitas:

$$C_{11} \equiv \left\{ \begin{array}{l} y^2 = x^3 + x^2 \\ x, y \in [-5, 5] \end{array} \right. \quad ; \quad C_{12} \equiv \left\{ \begin{array}{l} y^2 = x^3 \\ x, y \in [-5, 5] \end{array} \right.$$

$$C_{13} \equiv \begin{cases} (x^2+y^2)^2 = 2(x^2-y^2) \\ x, y \in [-5, 5] \end{cases}$$

$$C_{14} \equiv \begin{cases} (x^2+y^2)(x^2+y^2+3) + 16y^2(x^2+y^2-9) = 0 \\ x, y \in [-5, 5] \end{cases}$$

Adicionalmente, se pide calcular:

5. Los puntos de C_3 de ordenada $y = 2$.
6. El único punto en el primer cuadrante de la intersección $C_{13} \cap C_{14}$.
7. El valor máximo de la función de C_2.
8. Las soluciones reales de la ecuación $\frac{2}{3}x^4 + 8x^3 - 27x^2 + 18x + 27 = 0$.

2.2. Elementos necesarios

```
> with(plots):
```

```
Warning, the name changecoords has been redefined
```

2.3. Gráficas de funciones

2.3.1. Primera curva

```
> y:=16*(x-sin(x))/x^2;
> plot(y,x=0..2*Pi);
```

$$y := \frac{16\,(x - \sin(x))}{x^2}$$



2.3.2. Segunda curva

```
> y:=log(x)/x^3;
> plot(y,x=0.8..3);
```

$$y := \frac{\ln(x)}{x^3}$$



2.3.3. Tercera curva

```
> y:=exp(sin(x));
> plot(y,x=0..2*Pi);
```

$$y := e^{\sin(x)}$$



2.4. Coordenadas polares

```
> macro(rr=rho,aa=alpha);
```

$$rr,\ aa$$

Con esta orden cada vez que escribamos `rr`, `aa` es como si escribieramos ρ, α, respectivamente.

2.4.1. Cuarta curva: La circunferencia

```
> rr:=3;
> polarplot(rr,aa=0..1.9*Pi);
```

$$\rho := 3$$



2.4.2. Quinta curva: El trébol

```
> rr:=4*sin(3*aa);
> polarplot(rr,aa=0..2*Pi);
```

$$\rho := 4\sin(3\,\alpha)$$



2.4.3. Sexta curva: La Cardioide

```
> rr:=4*(1+cos(aa));
> polarplot(rr,aa=0..2*Pi);
```

$$\rho := 4 + 4\cos(\alpha)$$



2.4.4. Séptima curva: Elipse en polares

```
> rr:=1/(1+0.9*cos(aa));
> polarplot(rr,aa=0..(3/2)*Pi);
```

$$\rho := \frac{1}{1 + 0.9\cos(\alpha)}$$



2.4.5. Octava curva: La lemniscata

```
> rr:=sqrt(2*cos(2*aa));
> polarplot(rr,aa=0..2*Pi);
```

$$\rho := \sqrt{2}\,\sqrt{\cos(2\,\alpha)}$$



2.5. Curvas en paramétricas

2.5.1. Novena curva: La Astroide

```
> x:=cos(t)^3; y:=sin(t)^3;
> plot([x,y,t=0..2*Pi]);
```

$$x := \cos(t)^3$$

$$y := \sin(t)^3$$



2.5.2. Décima curva: La Cicloide

```
> x:=4*(t-sin(t)); y:=4*(1-cos(t));
> plot([x,y,t=0..4*Pi]);
```

$$x := 4\,t - 4\sin(t)$$
$$y := 4 - 4\cos(t)$$



2.6. Curvas en implícitas

Deshacemos las asignaciones anteriores sobre las variables x, y mediante el listado:

```
> x:='x': y:='y':
```

2.6.1. Undécima curva: El nodo

```
> Ec:=y^2=x^3+x^2;
> implicitplot(Ec,x=-5..5,y=-5..5,numpoints=5000);
```

$$Ec := y^2 = x^3 + x^2$$



2.6.2. Duodécima curva: La cúspide

```
> Ec:=y^2=x^3;
> implicitplot(Ec,x=-5..5,y=-5..5,numpoints=5000);
```

$$Ec := y^2 = x^3$$



2.6.3. Decimotercera curva: La lemniscata

```
> Ec13:=(x^2+y^2)^2=2*(x^2-y^2);
> G13:=implicitplot(Ec13,x=-5..5,y=-5..5,numpoints=10000):
> display(G13);
```

$$Ec13 := (x^2 + y^2)^2 = 2\,x^2 - 2\,y^2$$



2.6.4. Decimocuarta curva: curva de Watt

```
> Ec14:=(x^2+y^2)*(x^2+y^2+3)^2+16*y^2*(x^2+y^2-9)=0;
> G14:=implicitplot(Ec14,x=-5..5,y=-5..5,numpoints=10000):
> display(G14);
```

$$Ec14 := (x^2+y^2)\,(x^2+y^2+3)^2 + 16\,y^2\,(x^2+y^2-9) = 0$$



2.7. Cálculo de extremos y resolución de sistemas no lineales

2.7.1. Ceros de funciones. Resolución del punto 5

Calculamos el punto 5, lo que equivale a obtener los ceros de la función $f(x) = e^{\operatorname{sen} x} - 2$:

```
> f:=exp(sin(x))-2;
```

$$f := e^{\sin(x)} - 2$$

```
> c1:=fsolve(f=0,x=0..2);
> c2:=fsolve(f=0,x=2..4);
```

$$c1 := 0.7658461948$$
$$c2 := 2.375746459$$

Utilizando el método de Newton que explicamos en el tema 5, para calcular c_1 podemos poner:

```
> fder:=diff(f,x);
> F:=unapply(x-f/fder,x);
```

$$fder := \cos(x)\,e^{\sin(x)}$$

$$F := x \to x - \frac{e^{\sin(x)} - 2}{\cos(x)\,e^{\sin(x)}}$$

```
> c:=0.; c:=F(c); c:=F(c); c:=F(c); c:=F(c); c:=F(c);
```

$$c := 0.$$
$$c := 1.000000000$$
$$c := 0.7448685721$$
$$c := 0.7658012177$$

$$c := 0.7658461946$$
$$c := 0.7658461946$$

2.7.2. Resolución de sistemas no lineales. Resolución del punto 6

```
> display(G13,G14);
```




```
> c1:=fsolve({Ec13,Ec14},{x,y},{x=1..5,y=0..5});
```

$$c1 := \{x = 1.076074024,\ y = 0.4638897374\}$$

Con la primera instrucción pintamos las curvas 13^a y 14^a a la vez, además, si pinchamos sobre la gráfica en la única intersección del primer cuadrante obtenemos en el recuadro superior izquierdo una aproximación a dicha intersección, en concreto $1.07, 0.47$. La exactitud la conseguimos con la segunda instrucción.

2.7.3. Cálculo de extremos. Resolución del punto 7

```
> f:=log(x)/x^3; fder:= diff(f,x);
> Extremo:=fsolve(fder=0,x);
> ValorExactoDelExtremo:=solve(fder=0);
```

$$f := \frac{\ln(x)}{x^3}$$
$$fder := \frac{1}{x^4} - \frac{3\ln(x)}{x^4}$$
$$Extremo := 1.395612425$$
$$ValorExactoDelExtremo := e^{(1/3)}$$

Podemos apreciar la diferencia entre los comandos `fsolve` y `solve`; el segundo resuelve, siempre que puede, de forma exacta.

2.7.4. Factorización de polinomios. Resolución del punto 8

```
> P:=2/3*X^4+8*X^3-27*X^2+18*X+27;
```

$$P := \frac{2}{3}X^4 + 8X^3 - 27X^2 + 18X + 27$$

```
> factor(P,complex);
```

$$.6666666667(X + 14.83943212)(X + .6784947867)(X - 1.758963454 + .9635885359\,I)$$
$$(X - 1.758963454 - .9635885359\,I)$$

Deduciéndose que las soluciones reales de la ecuación son $x_1 = -0.6784947867$ y $x_2 = -14.83943212$.